
From Qᵢ = λᵢF + εᵢ to Σ = ΛΛ′ + Ψ
At this point, we have established the measurement model (Section 2.1), derived how a latent factor leaves its fingerprint in the covariance structure (Sections 2.2-2.3), and expressed the full result in matrix form (Section 2.4). The purpose of this section is to bring everything together in one place.
We will walk through a complete worked example from start to finish: simulating data with a known underlying structure, discarding the hidden mechanism, and then recovering it by hand using only the observable correlations. At the end, we verify the result against Stata's built-in estimator.
The goal is not to introduce a new technique. It is to make the whole derivation visible as one coherent object.
After completing this section, you should be able to:
explain how a latent factor generates observable correlations;
derive the covariance equation Cov(Qᵢ,Qⱼ)=λᵢλⱼ;
derive the variance decomposition Var(Qᵢ)=λᵢ²+ψᵢ;
explain why Σ=ΛΛ′ + Ψ is a compact representation of many covariance equations;
reproduce a one-factor solution manually; and
verify the derivation using Stata.
Step 0. The Setup: A Survey of Statistical Anxiety
Suppose we are designing a short questionnaire to measure statistical anxiety, an unobservable psychological disposition. We write three items:
	Item
	Wording

	q₁
	"Statistics formulas make me nervous."

	q₂
	"I worry that I cannot learn statistics."

	q₃
	"I feel anxious when doing statistics problems."



We assume that all three items are caused by a single latent factor F, interpreted as anxiety. The measurement model is
Qᵢ = λᵢF + εᵢ,    i = 1, 2, 3.
The true parameter values, known to us as data generators but unknown to the analyst, are
λ₁ = 0.83,    λ₂ = 0.87,    λ₃ = 0.78,
and
ψ₁ = 0.31,    ψ₂ = 0.24,    ψ₃ = 0.39.
Here λᵢ is the loading for item i, and ψᵢ=Var(εᵢ) is its unique variance.
Step 1. Simulate Data from the Known Model
We generate N=100 observations. First, we draw the latent factor
F ∼ N(0, 1),
which satisfies the normalization assumption Var(F)=1. Then we draw independent errors
εᵢ ∼ N(0, ψᵢ),
ensuring Cov(F,εᵢ)=0 and Cov(εᵢ,εⱼ)=0 for i≠ j.
clear all
set seed 2026
set obs 100

* Latent factor: Var(F) = 1
gen F = rnormal(0, 1)

* Unique errors: independent of F and of each other
gen epsilon1 = rnormal(0, sqrt(0.31))
gen epsilon2 = rnormal(0, sqrt(0.24))
gen epsilon3 = rnormal(0, sqrt(0.39))

* Observed variables: q_i = lambda_i * F + epsilon_i
gen q1 = 0.83 * F + epsilon1
gen q2 = 0.87 * F + epsilon2
gen q3 = 0.78 * F + epsilon3
We can verify the three assumptions directly in the simulated data:
* Assumption 1: Cov(F, epsilon_i) is approximately 0
correlate F epsilon1 epsilon2 epsilon3, covariance

* Assumption 2: Cov(epsilon_i, epsilon_j) is approximately 0 for i != j
correlate epsilon1 epsilon2 epsilon3, covariance

* Assumption 3: Var(F) is approximately 1
summarize F
Now we discard F and all εᵢ. From this point on, we pretend we are the analyst: we see only q₁,q₂,q₃.
drop F epsilon1 epsilon2 epsilon3
Step 2. Compute the Sample Covariance Matrix S
We standardize the three items so that all variances equal 1. Once the variables are standardized, the covariance matrix is the correlation matrix. This is standard practice in exploratory factor analysis when items are measured on comparable but not identical scales.
foreach var of varlist q1 q2 q3 {
    summarize `var'
    gen z_`var' = (`var' - r(mean)) / r(sd)
}

correlate z_q1 z_q2 z_q3
matrix S = r(C)
The resulting sample correlation matrix is
S =
⎛ 1       0.677   0.485 ⎞
⎜ 0.677   1       0.608 ⎟
⎝ 0.485   0.608   1     ⎠
This is now all we have to work with. The task of factor analysis is to find Λ and Ψ such that
ΛΛ′ + Ψ ≈ S.
The latent factor F has been discarded. We must recover its structure from the correlations alone.
Step 3. Estimate the Loadings λ from Off-Diagonal Elements
Recall from Section 2.2 that, under the three assumptions,
Cov(qᵢ, qⱼ) = λᵢλⱼ,    i ≠ j.
Because a one-factor model with three standardized variables contains exactly three unknown loadings and three off-diagonal correlations, the system is exactly identified. This allows us to solve the loadings algebraically, without iterative optimization.
The three equations are
r₁₂ = λ₁λ₂,    r₁₃ = λ₁λ₃,    r₂₃ = λ₂λ₃.
Divide the first equation by the third:
r₁₂ / r₂₃ = (λ₁λ₂) / (λ₂λ₃) = λ₁ / λ₃.
Similarly,
r₁₂ / r₁₃ = λ₂ / λ₃.
Substitute this relation back into r₂₃=λ₂λ₃:
r₂₃ = [(r₁₂ / r₁₃)λ₃]λ₃ = (r₁₂ / r₁₃)λ₃².
Solving for the third loading gives
λ₃ = √[(r₂₃r₁₃) / r₁₂].
The remaining two loadings follow immediately:
λ₁ = r₁₃ / λ₃,    λ₂ = r₂₃ / λ₃.
Equivalently, using the ratios above,
λ₁ = (r₁₂ / r₂₃)λ₃,    λ₂ = (r₁₂ / r₁₃)λ₃.
In Stata:
local r12 = S[1,2]    // 0.6766
local r13 = S[1,3]    // 0.4848
local r23 = S[2,3]    // 0.6080

local ratio_1 = `r12' / `r23'    // lambda_1 / lambda_3
local ratio_2 = `r12' / `r13'    // lambda_2 / lambda_3

local lambda3 = sqrt(`r23' / `ratio_2')
local lambda1 = `ratio_1' * `lambda3'
local lambda2 = `ratio_2' * `lambda3'
The estimated loadings are:
	Variable
	Manual λ̂
	True λ

	q₁
	0.7345
	0.83

	q₂
	0.9212
	0.87

	q₃
	0.6601
	0.78



The estimates deviate somewhat from the true values. This is expected: with N=100 and a purely algebraic method, sampling variability matters. But the ordering and relative magnitudes are preserved correctly.
Step 4. Estimate the Uniqueness ψ from Diagonal Elements
Recall from Section 2.3 that
Var(qᵢ) = λᵢ² + ψᵢ.
Since the variables are standardized, Var(qᵢ)=1 for every item. Therefore
ψᵢ = 1 - λᵢ².
The quantity λᵢ² is called the communality of item i: the proportion of the item's variance explained by the latent factor.
local psi1 = 1 - `lambda1'^2
local psi2 = 1 - `lambda2'^2
local psi3 = 1 - `lambda3'^2
	Variable
	λ̂
	Communality λ̂²
	Uniqueness ψ̂

	q₁
	0.7345
	0.5394
	0.4606

	q₂
	0.9212
	0.8485
	0.1515

	q₃
	0.6601
	0.4357
	0.5643



Interpretation. Item q₂, "I worry that I cannot learn statistics," has the highest communality, approximately 0.85. Nearly all of its variance is accounted for by the latent anxiety factor, making it the purest indicator. Item q₃ has the most unique variance, which means it appears to capture something that the common factor alone does not fully explain.
Step 5. Reconstruct Σ̂=ΛΛ′ + Ψ
We now verify that our estimated parameters reproduce the original correlation matrix. Construct the loading vector and uniqueness matrix:
Λ̂ =
⎛ 0.7345 ⎞
⎜ 0.9212 ⎟
⎝ 0.6601 ⎠

Ψ̂ =
⎛ 0.4606   0        0      ⎞
⎜ 0        0.1515   0      ⎟
⎝ 0        0        0.5643 ⎠
In Stata:
matrix LAMBDA = [`lambda1' \ `lambda2' \ `lambda3']
matrix PSI    = diag((`psi1', `psi2', `psi3'))

matrix SIGMA_HAT = LAMBDA * LAMBDA' + PSI
matrix RESIDUAL  = S - SIGMA_HAT

matrix list RESIDUAL
The residual matrix is
S - Σ̂ ≈
⎛ 0   -4.4×10⁻¹⁶   -1.7×10⁻¹⁶ ⎞
⎜ ·    0             -1.1×10⁻¹⁶ ⎟
⎝ ·    ·              0          ⎠
The off-diagonal residuals are numerically zero; the small values are only floating-point rounding. This confirms that the algebraic solution exactly reproduces S. In real data, non-zero residuals indicate sampling variability, model misspecification, or too few factors.
Step 6. Verify Against Stata's Built-In Estimator
Stata's factor command uses iterative principal factors (IPF), a more sophisticated algorithm than our algebraic solution. We compare:
factor z_q1 z_q2 z_q3, ipf factors(1)
Stata's output is:
Factor loadings (pattern matrix) and unique variances

    Variable |  Factor1 |   Uniqueness
    ---------+----------+--------------
        z_q1 |   0.7346 |      0.4604
        z_q2 |   0.9210 |      0.1518
        z_q3 |   0.6601 |      0.5642
Comparing Stata to the manual estimates:
	Parameter
	Manual
	Stata IPF

	λ̂₁
	0.7345
	0.7346

	λ̂₂
	0.9212
	0.9210

	λ̂₃
	0.6601
	0.6601

	ψ̂₁
	0.4606
	0.4604

	ψ̂₂
	0.1515
	0.1518

	ψ̂₃
	0.5643
	0.5642



The results are virtually identical. The small differences arise because Stata's IPF algorithm iterates to a refined solution, while our algebraic approach solves the exactly identified system directly. The underlying logic is the same: find the loading matrix Λ and uniqueness matrix Ψ that reproduce the observed covariance structure.
Step 7. Computing Factor Scores
Once loadings are estimated, we can compute a factor score for each respondent: a weighted composite of responses that serves as the best reconstruction of the respondent's underlying anxiety level.
predict factor_score
label variable factor_score "Statistical Anxiety (Factor Score)"

summarize factor_score
The scoring coefficients printed by Stata indicate the weight assigned to each item:
	Item
	Scoring weight

	z_q₁
	0.187

	z_q₂
	0.711

	z_q₃
	0.137



Notice that q₂, the item with the highest loading, receives the greatest weight. This is not arbitrary. Items that are more strongly connected to the latent factor carry more information about it, and the regression-based scoring method formalizes this intuition.
Factor scores are standardized: F̄≈0 and SD(F)≈1. A score of +1 indicates a respondent one standard deviation above average in statistical anxiety; -1 indicates one standard deviation below average.
Summary: The Complete Logic Chain
The derivation in this section follows a single continuous thread:
Qᵢ = λᵢF + εᵢ
  assumptions ⟶ Cov(qᵢ, qⱼ) = λᵢλⱼ
  solve ⟶ λ̂ᵢ
The diagonal elements provide the second part:
Var(qᵢ) = λᵢ² + ψᵢ
  solve ⟶ ψ̂ᵢ = 1 - λ̂ᵢ²
  combine ⟶ Σ̂ = Λ̂Λ̂′ + Ψ̂ ≈ S
The matrix equation
ΛΛ′ + Ψ ≈ S.
is not a new model. It is simply a compact notation for the covariance equations derived from the measurement model. Every element in that matrix corresponds to a statement about how observable variables co-vary because they share a common cause.
Core insight. The latent factor F is unobservable, but it is not undetectable. Its presence leaves a systematic fingerprint in the off-diagonal elements of the correlation matrix. Factor analysis is the method for reading that fingerprint: recovering the hidden structure that generated the correlations we observe.

